
Abstract

The TPTP language, developed within the framework of the TPTP library, allows the
representation of problems and solutions in first-order and higher-order logic. Whereas
the writing of solutions in resolution calculi is well documented and used, an appropriate
representation of solutions in tableau or connection calculi using the TPTP syntax has not
yet been specified. This paper describes how the TPTP language can be used to represent
derivations and solutions in standard tableau, sequent and connection calculi for classical
first-order logic.

1 Introduction

The TPTP language specifies syntax and semantics for expressing problems in first-order and
higher-order logic. It is used not only within the TPTP library [23], but also within similar
problem libraries, e.g., the ILTP library [15]. The TPTP syntax for representing problems
is used by a variety of automated theorem proving (ATP) systems based on different proof
calculi. The TPTP language also allows representation of solutions, e.g., derivations and models,
produced by ATP systems. The writing of derivations in resolution calculi is well documented
and specified [25]. At the last CADE system competition, CASC-22 [24], three of the five ATP
systems that output proofs in the core FOF division use the TPTP syntax. All three of those
systems produce proofs that are based on resolution calculi.

Even though the TPTP syntax is flexible, the presentation of derivations in, e.g., tableau,
sequent or connection calculi is not straightforward. Derivations in these calculi differ signifi-
cantly from derivations in the resolution calculus. Whereas the leaves of a proof in the tableau
calculus consists of the axioms of the calculus, the leaves of a derivation in the resolution calcu-
lus consists of the formulae of the given problem; the axiom of the (formal) resolution calculus
is the empty clause [18], which occurs only at the root of a refutation.

This paper describes how the TPTP language can be used to represent derivations and
proofs in standard tableau and connection calculi. As the sequent calculus is closely related to
the tableau calculus, this can easily be adapted to present derivations in the sequent calculus as
well. This is a proposed format, not yet formally established as a TPTP standard; community
feedback with suggestions for improvement are welcome. The goal is to produce a format that
is compatible with the existing format for representing derivations (reviewed in Section 2.2), so
that existing TPTP infrastructure for proof processing, e.g., the GDV proof verifier [21] and
the IDV proof visualizer [26], can be used with little or no modification.

2 The TPTP Language

The TPTP language is suitable for representing problems as well as derivations in first-order and
higher-order logic. The following description presents its main concepts. A detailed definition
is part of the TPTP library [23]; see also [25].

2.1 Representing Problems

The top level building blocks for problems using the TPTP syntax are annotated formulae, of
the following form:

language(name,role,formula,source,useful info).

The language is one of thf, fof, or cnf, for formulae in typed higher-order, first-order, and
clause normal form. Each annotated formula has a unique name. The role is, e.g., axiom

or conjecture. The source describes where the formula came from, e.g., an input file, and
useful info is a list of user information. The last two fields are optional.

Example 1. Pelletier’s problem 24 [14] consists of the following subformulae.

¬(∃x(Sx ∧Qx)) Axiom 1 (1)
∀(Px⇒ (Qx ∨Rx)) Axiom 2 (2)
¬(∃xPx)⇒ ∃yQy Axiom 3 (3)
∀x((Qx ∨Rx)⇒ Sx) Axiom 4 (4)
∃x(Px ∧Rx) Conjecture (5)

It stands for the first-order formula (Axiom 1∧Axiom 2∧Axiom 3∧Axiom 4)⇒ Conjecture .
This problem is in the TPTP library under the name SYN054+1. Its representation using the
TPTP syntax is given in Figure 1 (with an abbreviated version of the full TPTP header).

2.2 Representing Derivations

A derivation (in the resolution calculus) is a directed acyclic graph whose leaf nodes are formulae
from the problem, and whose interior nodes are formulae inferred from parent formulae. A
refutation is a derivation that has the root node false, representing the empty clause. A

%------------------------------------------------------------------------

% File : SYN054+1 : TPTP v4.0.1. Released v2.0.0.

% Domain : Syntactic

% Problem : Pelletier Problem 24

% Status : Theorem

% Rating : 0.00 v2.1.0

%------------------------------------------------------------------------

fof(pel24_1,axiom, ( ~ ( ? [X] : ( big_s(X) & big_q(X) ) ) )).

fof(pel24_2,axiom, ( ! [X] : ( big_p(X) => ( big_q(X) | big_r(X) ) ) )).

fof(pel24_3,axiom, ( ~ ( ? [X] : big_p(X) ) => ? [Y] : big_q(Y) )).

fof(pel24_4,axiom, ( ! [X] : ( ( big_q(X) | big_r(X) ) => big_s(X) ) )).

fof(pel24,conjecture, ( ? [X] : ( big_p(X) & big_r(X) ) )).

%------------------------------------------------------------------------

Figure 1: The presentation of the TPTP problem SYN054+1

∀(Px⇒ (Qx ∨Rx)) (2)

∃x(Px ∧Rx) (5)

¬(∃x(Sx ∧Qx)) (1)

∀x((Qx ∨Rx)⇒ Sx) (4)

{¬Pz,Qz,Rz} (12) {¬Px,¬Rx} (24) {¬Sy,¬Qy} (9) {¬Qv, Sv} (21)

Figure 2: A derivation for SYN054+1 in the resolution calculus

derivation written in the TPTP language is a list of annotated formulae, as for problems. For
derivations the source has one of the forms

file(file name,file info)

inference(inference name,inference info,parents)

The former is used for formulae taken from the problem file. The latter is used for inferred
formulae, in which inference name is the name of the inference rule applied by the ATP system,
inference info is a []ed list of additional information about the inference, and parents is a list
of the (logical) parents’ node names in the derivation. The inference info normally includes
a status() term that record the semantic relationship between the parents and the inferred
formula as an SZS ontology value [22]. Variable bindings applied to a logical parent are captured
in bind/2 terms following the parent’s name.

Example 2. Figure 2 shows a conversion of some of the axioms and the negated conjecture
of problem SYN054+1 from Example 1 to clause normal form, and a subsequent refutation of
the clause normal form in the resolution calculus [16]. The leaf nodes are the formulae of the
problem. As the root node is the empty clause, the derivation is a proof for problem SYN054+1.
The representation of this derivation using the TPTP syntax is given in Figure 3. It is a slightly
simplified version of the original proof output by the EP system [17]. The five inferences of the
resolution proof are represented by the nodes 25 to 29. The nodes of the proof in Figure 2 are
annotated by the corresponding EP node numbers.

3 Representing Derivations in the Tableau Calculus

Tableau calculi are well-known proof search calculi for classical and non-classical logics [4, 6].
The axiom and 12 rules of a standard tableau calculus for classical logic are given in Table 1 [20].
It uses signed formulae of the form AT or AF , in which A is a first-order formula and T/F is its
sign (or polarity). The signed formula AF can be interpreted as A⇒ false. The usage of signed
formulae allows an elegant and uniform representation of the rules of the tableau calculus. The
α-rules add formulae to a branch of a derivation, and the β-rules split a branch of the derivation

%--------------------------------------------------------------------------------------------
fof(1, axiom,~(?[X1]:(big_s(X1)&big_q(X1))),file(’SYN054+1.p’,pel24_1)).
fof(2, axiom,![X1]:(big_p(X1)=>(big_q(X1)|big_r(X1))),file(’SYN054+1.p’,pel24_2)).
fof(3, axiom,(~(?[X1]:big_p(X1))=>?[X2]:big_q(X2)),file(’SYN054+1.p’,pel24_3)).
fof(4, axiom,![X1]:((big_q(X1)|big_r(X1))=>big_s(X1)),file(’SYN054+1.p’,pel24_4)).
fof(5, conjecture,?[X1]:(big_p(X1)&big_r(X1)),file(’SYN054+1.p’,pel24)).
fof(6, negated_conjecture,~(?[X1]:(big_p(X1)&big_r(X1))),inference(assume_negation,[],[5])).
fof(7, plain,![X1]:(~(big_s(X1))|~(big_q(X1))),inference(fof_nnf,[],[1])).
fof(8, plain,![X2]:(~(big_s(X2))|~(big_q(X2))),inference(variable_rename,[],[7])).
cnf(9, plain,(~big_q(X1)|~big_s(X1)),inference(split_conjunct,[],[8])).
fof(10,plain,![X1]:(~(big_p(X1))|(big_q(X1)|big_r(X1))),inference(fof_nnf,[],[2])).
fof(11,plain,![X2]:(~(big_p(X2))|(big_q(X2)|big_r(X2))),inference(variable_rename,[],[10])).
cnf(12,plain,(big_r(X1)|big_q(X1)|~big_p(X1)),inference(split_conjunct,[],[11])).
fof(13,plain,(?[X1]:big_p(X1)|?[X2]:big_q(X2)),inference(fof_nnf,[],[3])).
fof(14,plain,(?[X3]:big_p(X3)|?[X4]:big_q(X4)),inference(variable_rename,[],[13])).
fof(15,plain,(big_p(esk1_0)|big_q(esk2_0)),inference(skolemize,[],[14])).
cnf(16,plain,(big_q(esk2_0)|big_p(esk1_0)),inference(split_conjunct,[],[15])).
fof(17,plain,![X1]:((~(big_q(X1))&~(big_r(X1)))|big_s(X1)),inference(fof_nnf,[],[4])).
fof(18,plain,

![X2]:((~(big_q(X2))&~(big_r(X2)))|big_s(X2)),inference(variable_rename,[],[17])).
fof(19,plain,

![X2]:((~(big_q(X2))|big_s(X2))&(~(big_r(X2))|big_s(X2))),inference(distribute,[],[18])).
cnf(21,plain,(big_s(X1)|~big_q(X1)),inference(split_conjunct,[],[19])).
fof(22,negated_conjecture,![X1]:(~(big_p(X1))|~(big_r(X1))),inference(fof_nnf,[],[6])).
fof(23,negated_conjecture,

![X2]:(~(big_p(X2))|~(big_r(X2))),inference(variable_rename,[],[22])).
cnf(24,negated_conjecture,(~big_r(X1)|~big_p(X1)),inference(split_conjunct,[],[23])).

cnf(25,plain,(big_q(X1)|~big_p(X1)),inference(csr,[],[12,24])).
cnf(26,plain,(~big_q(X1)),inference(csr,[],[9,21])).
cnf(27,plain,(big_p(esk1_0)),inference(sr,[],[16,26])).
cnf(28,plain,(~big_p(X1)),inference(sr,[],[25,26])).
cnf(29,plain,($false),inference(sr,[],[27,28])).
%--------------------------------------------------------------------------------------------

Figure 3: A derivation for SYN054+1 in the resolution calculus using the TPTP syntax

into two branches. When eliminating a universal quantifier using the γ-rule, all free occurrences
of the variable x in A are replaced by the term t. In the δ-rule the variable x is replaced by a
Skolem term that consists of a unique Skolem function symbol ski and all variables x1, . . . , xn
that occur free in A. A formula A is valid if, and only if, there is a derivation of AF in the
tableau calculus.

Example 3. A derivation of problem SYN054+1 from Example 1 in the tableau calculus is
shown in Figure 4. It follows the common representation of standard tableau calculi for classical
logic [6]. Each node is annotated in ( )s by its number and in [ ]s by the number of the node
whose formula is used as the premise of the inference rule, its logical parent. Additionally, a
substitution is given when the γ- or δ-rule is applied. The constants a and b are Skolem terms.
Branches that are closed by an axiom are marked with ×. The derivation in Figure 4 is not a
proof, because the rightmost branch (node 25) is not closed by an axiom.

Even though a derivation in the tableau calculus is still an acyclic directed graph, its struc-
ture is different from the structure of a derivation in the resolution calculus. Hence the proof
presentation using the TPTP language needs to be adapted. For a tableau, in addition to
listing the logical parents of each formula in the parents list, the physical parent of each node is
recorded in a pparent() term in the inference information list. The branching of the tableau is
recorded in the same way as splitting inferences are recorded in CNF refutations [25, 21]. The
inference name is axiom or the name of the applied inference rule, i.e., and T, or F, implies F,
neg T, neg F, and F, or T, implies T, forall T, exists F, forall F, or exists T. A formula
of the form AF is represented in the TPTP language by the formula A=>$false, if A is a

98



TPTP Derivations in Tableau and Connection Calculi J. Otten, G. Sutcliffe

Table 1: The axiom and the rules of the tableau calculus

Axiom AT

AF

⊥

α-rules ∧T (A ∧B)T

AT

BT

∨F (A ∨B)F

AF

BF

⇒F (A⇒ B)F

AT ∧BF

¬T (¬A)T

AF
¬F (¬A)F

AT

β-rules ∧F (A ∧B)F

/\
AF BF

∨T (A ∨B)T

/\
AT BT

⇒T (A⇒ B)T

/\
AF BT

γ-rules ∀T (∀xA)T

AT [x\t]
∃F (∃xA)F

AF [x\t]

δ-rules ∀F (∀xA)F

AF [x\ski(x1, ..., xn)]
∃T (∃xA)T

AT [x\ski(x1, ..., xn)]

non-atomic formula; otherwise, if A is an atomic formula, it is represented by ~A. A formula
of the form AT is represented by A.

Example 4. The derivation of Figure 4 is shown using the TPTP syntax in Figure 5. The
non-negated original conjecture is added as node 0. Observe the fact that the physical parent
might differ from the logical parent. For example, the physical parent of node 8 is node 7,
its formula is (¬(∃xPx))F , which is obtained by an implies T inference, whose premise is the
formula of node 4.

The proof representation is independent from the specific proof search (algorithm). Hence,
the proposed format can also be used to represent derivations obtained by, e.g., free-variable
tableaux or a proof search using iterative deepening.

Representing Derivations in the Sequent Calculus. The tableau calculus is closely
related to the sequent calculus [5]. Therefore, the TPTP format for tableau derivations can
also be used for representing derivations in standard sequent calculi [20]. Formulae of the
form AT occur (only) on the left side of the sequents (the antecedent), formulae of the form
AF occur (only) on the right side (the succedent). Each inference rule ruleT and ruleF in the
tableau calculus corresponds to exactly one rule rule-left and rule-right in the sequent calculus,
respectively.

Example 5. Figure 6 shows the the sequent derivation that corresponds to the tableau derivation
given in Figure 4. However, as the Eigenvariable condition needs to be respected (for the ∃-left∗

rule), it might be necessary to reorder some inference rules in order to obtain a correct proof.

(∃x(Px ∧Rx))F (1)
|

(¬(∃x(Sx ∧Qx)))T (2)
|

(∀x(Px⇒ (Qx ∨Rx)))T (3)
|

(¬(∃xPx)⇒ ∃yQy)T (4)
|

(∀x((Qx ∨Rx)⇒ Sx))T (5)
|

(Pa ∧Ra)F (6) [1, {x\a}]
���

���

XXXXXXXXXXX
PaF (7) [6] RaF (25) [6]
���

���

XXXXXXXXXXX
(¬(∃xPx))F (8) [4]

|
(∃xPx)T (9) [8]
|

PaT (10) [9, {x\a}]
|
× (11) [7, 10]

(∃yQy)T (12) [4]
|

QbT (13) [12, {y\b}]
|

(∃x(Sx ∧Qx))F (14) [2]
|

(Sb ∧Qb)F (15) [14, {x\b}]
�����������

H
HHH

HH

SbF (16) [15]
|

((Qb ∨Rb)⇒ Sb)T (17) [5, {x\b}]

QbF (23) [15]
|
× (24) [13, 23]

���
���

XXXXXXXXXXX

(Qb ∨Rb)F (18) [17]
|

QbF (19) [18]
|
× (20) [13, 19]

SbT (21) [17]
|
× (22) [16, 21]

Figure 4: A derivation for SYN054+1 in the tableau calculus

4 Representing Derivations in the Connection Calculus

Connection calculi, e.g. the connection method [2], the connection tableau calculus [8] and the
model elimination calculus [9], are established proof search calculi. In principle, derivations
in the clausal connection calculus can be seen as derivations in the tableau calculus with a
connectedness condition [6]. But to this end many additional inferences need to be inserted.

%------------------------------------------------------------------------------------------------------
fof(0, conjecture,?[X]:(big_p(X)&big_r(X)),file(’SYN054+1.p’,pel24)).
fof(1, negated_conjecture,(~ ?[X]:(big_p(X)&big_r(X)))=>$false,inference(neg_conj,[pparent([0])],[0])).
fof(2, axiom,~ ?[X]:(big_s(X)&big_q(X)),file(’SYN054+1.p’,pel24_1)).
fof(3, axiom,![X]:(big_p(X)=>(big_q(X)|big_r(X))),file(’SYN054+1.p’,pel24_2)).
fof(4, axiom,~ ?[X]:big_p(X)=>?[Y]:big_q(Y),file(’SYN054+1.p’,pel24_3)).
fof(5, axiom,![X]:((big_q(X)|big_r(X))=>big_s(X)),file(’SYN054+1.p’,pel24_4)).

fof(6, plain,(big_p(X)&big_r(X))=>$false,
inference(exists_F,[status(thm),pparent([5])],[1:[bind(X,$fot(a))]])).

fof(7, plain,~big_p(a),inference(and_F,[and_F(split,[position(l)]),pparent([6])],[6])).
fof(8, plain,(~ ?[X]:big_p(X))=>$false,

inference(implies_T,[implies_T(split,[position(ll)]),pparent([7])],[4])).
fof(9, plain,?[X]:big_p(X),inference(neg_F,[status(thm),pparent([8])],[8])).
fof(10,plain,big_p(a),inference(exists_T,[status(thm),pparent([9])],[9])).
fof(11,plain,$false,inference(axiom,[status(thm),pparent([10])],[7,10])).

fof(12,plain,?[Y]:big_q(Y),inference(implies_T,[implies_T(split,[position(lr)]),pparent([7])],[4])).
fof(13,plain,big_q(b),inference(exists_T,[status(thm),pparent([12])],[12:[bind(Y,$fot(b))]])).
fof(14,plain,(?[X]:(big_s(X)&big_q(X)))=>$false,inference(neg_T,[status(thm),pparent([13])],[2])).
fof(15,plain,(big_s(b)&big_q(b))=>$false,

inference(exists_F,[status(thm),pparent([14])],[14:[bind(X,$fot(b))]])).

fof(16,plain,~big_s(b),inference(and_F,[and_F(split,[position(lrl)]),pparent([15])],[15])).
fof(17,plain,(big_q(b)|big_r(b))=>big_s(b),

inference(forall_T,[status(thm),pparent([16])],[5:[bind(X,$fot(b))]])).
fof(18,plain,(big_q(b)|big_r(b))=>$false,

inference(implies_T,[implies_T(split,[position(lrll)]),pparent([17])],[17]).
fof(19,plain,~big_q(b),inference(or_F,[status(thm),pparent([18])],[18])).
fof(20,plain,$false,inference(axiom,[status(thm),pparent([19])],[13,19])).
fof(21,plain,big_s(b),inference(implies_T,[implies_T(split,[position(lrlr)]),pparent([17])],[17])).
fof(22,plain,$false,inference(axiom,[status(thm),pparent([21])],[16,21])).

fof(23,plain,~big_q(b),inference(and_F,[and_F(split,[position(lrr)]),pparent([15])],[15])).
fof(24,plain,$false,inference(axiom,[status(thm),pparent([23])],[13,23])).
fof(25,plain,~big_r(a),inference(and_F,[and_F(split,[position(r)]),pparent([6])],[6])).
%------------------------------------------------------------------------------------------------------

Figure 5: A derivation for SYN054+1 in the tableau calculus using the TPTP syntax

Hence it is advantageous to have a different representation of derivations, in which each inference
in the connection calculus is related to exactly one inference in the representation (using the
TPTP language).

(11)

(2), (3), (5),Pa ` Pa (10)
axiom

(2), (3), (5), (∃xPx) ` Pa (9)
∃-left∗

(2), (3), (5) ` Pa,¬(∃xPx) (8)
¬-right

(20)

(3),Qb ` Qb, Rb, Sb, Pa (19)
axiom

(3), Qb ` (Qb ∨ Rb), Sb, Pa (18)
∨-right

(22)

(3),Sb, Qb ` Sb, Pa (21)
axiom

(3), (Qb ∨ Rb)⇒ Sb,Qb ` Sb, Pa (17)
⇒-left

(3), ∀x((Qx ∨ Rx)⇒ Sx), Qb ` Sb, Pa (16)
∀-left

(24)

(3), (5),Qb ` Qb, Pa (23)
axiom

(3), (5), Qb ` Sb ∧Qb, Pa (15)
∧-right

(3), (5), Qb ` ∃x(Sx ∧Qx), Pa (14)
∃-right

¬(∃x(Sx ∧Qx)), (3), (5), Qb ` Pa (13)
¬-left

(2), (3), (5), ∃yQy ` Pa (12)
∃-left∗

(2), (3),¬(∃xPx)⇒ ∃yQy, (5) ` Pa (7)
⇒-left

. . . ` Ra (25)

(2) ∧ (3) ∧ (4) ∧ (5) ` Pa ∧ Ra (6)
∧-right

(2), (3), (4), (5) ` ∃x(Px ∧ Rx)
∃-right

Figure 6: A derivation for SYN054+1 in the sequent calculus

Table 2: The rules of the (clausal) connection calculus

Start rule ������
L1

�
�
�

L2

. . .
. . .

Z
Z
Z

. . .

PPPPPP
Ln (σ)

is a derivation for a clause C =
{L1, . . . , Ln} and a (term) sub-
stitution σ.

Reduction rule ...

D . . .
L (τ)

������
L1

�
�
�

L2

. . .
. . .

S
S
Li

. . .
. . .

PPPPPP
Ln (σ)

is a derivation, if D (without the
thick line) is a derivation, Li is
a (leaf) literal not element of a
connection, L is a literal on the
path from Li to the root, and
{τ(L), σ(Li)} is a connection.

Extension rule

�
�
�
�
�
�

Q
Q

Q
Q

Q
Q

D L (τ)
������

L1

�
�
�

L2

. . .
. . .

S
S
Li

. . .
. . .

PPPPPP
Ln (σ)

is a derivation for a clause C =
{L1, . . . , Ln} and a substitution
σ, if D is a derivation, L is a
(leaf) literal not element of a con-
nection, and {τ(L), σ(Li)} is a
connection for some i.

The main concept of connection calculi is the guidance of the proof search by connections. A
connection is a set of literals with opposite polarity but identical atomic formulae, i.e., {L1, L2}
is a connection if, and only if, L1 = ¬L2 or ¬L1 = L2. The connection calculus has three
main inference rules: start , reduction, and extension rule. These rules are depicted in Table 2.
For details see [2, 8, 11]. A formula F in disjunctive (conjunctive) clause normal form is valid
(unsatisfiable) if, and only if, there is a derivation in the (clausal) connection calculus such that
every literal is an element of at least one connection.

Example 6. A derivation of problem SYN054+1 from Example 1 in the clausal connection
calculus is shown in Figure 7 and Figure 8. Again, each inference is annotated by its number,
the clause number used in the inference, and a substitution. The inferences with numbers 5 and
10 are applications of reduction rules. The branch containing the circled literal, i.e., inference
number 7, is closed by an application of the lemma rule (see [11] for details). The derivation is
a proof as every literal is element of a connection, hence all branches are closed by at least one
connection. The major left branch in Figure 7 is a compact representation of the derivation
shown in Figure 4, containing only the bold literals of Figure 4.

A derivation in the clausal connection calculus using the TPTP language is a list of clausal
annotated formulae, as described in Section 2.2. Similar to a tableau, the parents is an ordered
list in which the first element is the name of the physical parent of the node, and the following
element is the name of the logical parent, i.e., the clause of the inference. Again, variable
bindings are captured in bind/2 terms. Additionally, the number of the selected literal of
the physical parent is captured in a cnf selected/1 term. Optionally, such a term can be
assigned to the logical parent as well, which would make it easier to identify the connection.
The inference name is the name of the applied inference rule, i.e., either start, reduction,
extension, or lemma.

```````````̀
¬Pa ¬Ra (1) [1, {x\a}]
�

��

�������
C
CC

HH

Pa Qb (2) [4] ¬Pa����
Qa Ra (6) [3, {z\a}]

�
��

@
@@

�
��

@
@@

¬Sb ¬Qb (3) [2, {y\b}]

(7)

¬Sa ¬Qa (8) [2, {y\a}]
�

��

@
@@

�
��

@
@@

¬Qb Sb (4) [5, {v\b}] ¬Qa Sa (9) [5, {v\a}]

(5) (10)

Figure 7: A derivation for SYN054+1 in the connection calculus (tableau representation)

Example 7. The derivation of Figure 7 is shown using this TPTP syntax in Figure 9. The
output is produced by the most recent version of the leanCoP system [12, 10].

5 Conclusion

A proposal for representing standard tableau, sequent and connection calculi in the TPTP
language has been presented. Even though derivations in these calculi differ significantly from
those in the resolution calculus, the existing TPTP syntax is flexible enough to represent deriva-
tions in these calculi as well. A common standard for presentation of derivations and proofs will
increase the interoperability between ATP systems, ATP tools, and application software (see,


¬Px
¬Rx

¬Sy
¬Qy

¬Pz
Qz

Rz

Pa

Qb

¬Qv

Sv

Sw

¬Rw




¬Px
¬Rx

¬Sy
¬Qy

¬Pz
Qz

Rz

Pa

Qb

¬Qv

Sv

Sw

¬Rw


����

(1) ↓
(2)

(3)

(4)

(5)

(6)

(7)

(8)

(9)

(10)

(1) {x\a}
(3) {y\b}
(4) {v\b}

(6) {z\a}
(8) {y\a}
(9) {v\a}

Figure 8: A derivation for SYN054+1 in the connection calculus (matrix representation)

%-----------------------------------------------------------------------------------------------
fof(pel24_1,axiom,~ ?[X]:(big_s(X)&big_q(X)),file(’SYN054+1.p’,pel24_1)).
fof(pel24_2,axiom,![X]:(big_p(X)=>(big_q(X)|big_r(X))),file(’SYN054+1.p’,pel24_2)).
fof(pel24_3,axiom,~ ?[X]:big_p(X)=>?[Y]:big_q(Y),file(’SYN054+1.p’,pel24_3)).
fof(pel24_4,axiom,![X]:((big_q(X)|big_r(X))=>big_s(X)),file(’SYN054+1.p’,pel24_4)).
fof(pel24,conjecture,?[X]:(big_p(X)&big_r(X)),file(’SYN054+1.p’,pel24)).

fof(f0,negated_conjecture,~ ?[X]:(big_p(X)&big_r(X)),
inference(negate_conjecture,[status(cth)],[pel24])).

cnf(c1,plain,(~big_p(X)|~big_r(X)),inference(clausify,[status(esa)],[f0])).
cnf(c2,plain,(~big_s(Y)|~big_q(Y)),inference(clausify,[status(esa)],[pel24_1])).
cnf(c3,plain,(~big_p(Z)|big_q(Z)|big_r(Z)),inference(clausify,[status(esa)],[pel24_2])).
cnf(c4,plain,(big_p(a)|big_q(b)),inference(clausify,[status(esa)],[pel24_3])).
cnf(c5,plain,(~big_q(V)|big_s(V)),inference(clausify,[status(esa)],[pel24_4])).

cnf(1,plain,(~big_p(a)|~big_r(a)),
inference(start,[status(thm)],[c1:[bind(X,$fot(a))]])).

cnf(2,plain,(big_p(a)|big_q(b)),
inference(extension,[status(thm),pparent([1:[cnf_select([1])]])],[c4])).

cnf(3,plain,(~big_s(b)|~big_q(b)),
inference(extension,[status(thm),pparent([2:[cnf_select([2])]])],[c2:[bind(Y,$fot(b))]])).

cnf(4,plain,(~big_q(b)|big_s(b)),
inference(extension,[status(thm),pparent([3:[cnf_select([1])]])],[c5:[bind(V,$fot(b))]])).

cnf(5,plain,$false,
inference(reduction,[pparent([4:[cnf_select([1])]])],[2])).

cnf(6,plain,(~big_p(a)|big_q(a)|big_r(a)),
inference(extension,[status(thm).pparent([1:[cnf_select([2])]])],[c3:[bind(Z,$fot(a))]])).

cnf(7,plain,$false,
inference(lemma,[pparent([6:[cnf_select([1])]])],[1])).

cnf(8,plain,(~big_s(a)|~big_q(a)),
inference(extension,[status(thm),pparent([6:[cnf_select([2])]])],[c2:[bind(Y,$fot(a))]])).

cnf(9,plain,(~big_q(a)|big_s(a)),
inference(extension,[status(thm),pparent([8:[cnf_select([1])]])],[c5:[bind(V,$fot(a))]])).

cnf(10,plain,$false,
inference(reduction,[pparent([9:[cnf_select([1])]])],[6])).

%-----------------------------------------------------------------------------------------------

Figure 9: A derivation for SYN054+1 in the connection calculus using the TPTP syntax

e.g., [19]). Future work includes the development of tools to translate connection proofs into
sequent proofs, which are often used in interactive proof editors, such as Coq [1], NuPRL [3] or
PVS [13]. A possible extension of the current work includes the use of the TPTP language to
represent derivations in tableau and connection calculi for non-classical logics, e.g., intuitionis-
tic and modal logics [7, 27]. These calculi often use additional annotations, e.g., a prefix that
is assigned to each formula. The TPTP syntax might need to be carefully extended in order to
allow the presentation of derivations in these calculi as well.

